/**
 * @(#) PostProxy.as
 *
 * @date		2009. 02. 17.
 * @version		1.0
 * @author		Shin, Yunwoo
 * @description	 우편번호 검색 - Proxy
 * 				
 */
package com.skt.common_hr.myprofile_stc.model
{
	import com.skt.common_hr.common.puremvc.CommonProxy;
	import com.skt.common_hr.myprofile_stc.MyProfileFacade;
	import com.skt.common_hr.myprofile_stc.model.vo.PostVO;
	
	import mx.collections.ArrayCollection;
	import mx.rpc.events.FaultEvent;
	import mx.rpc.events.ResultEvent;
	
	
	public class PostProxy extends CommonProxy
	{
		public static const NAME:String = "PostProxy";
		
		public var postList:ArrayCollection;


		/**
		 * 생성자
		 * 
		 * @param
		 * @return void
		 */
		public function PostProxy(data:Object=null)
		{
			super(NAME, data);
		}
		
		
		/**
		 * data load를 수행
		 * 		- DataLoader 를 생성하고 Import Parameter 객체를 생성, RFC를 호출
		 * 
		 * @param pernr 사번
		 * @return void
		 */
		public function load(keyword:String):void
		{
			var param:Object = new Object();
			param["I_TEXT"] = keyword;
			loadData("Z_HR_0009", param, result, fault); 
		}


		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Event Listeners
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 정상적으로 리턴된 결과를 처리
		 * 
		 * @param
		 * @return void
		 */
		public function result(event:ResultEvent):void
		{
			if(retCode == 0) {  
				postList = parsePost(event.result.IT_POST);
				facade.sendNotification(MyProfileFacade.LOAD_POST_COMPLETE, this);
			}
			else {
				facade.sendNotification(MyProfileFacade.LOAD_POST_FAILED, this);
			}
		}
		
		/**
		 * 서버에서 오류가 발생했을 처리
		 * 
		 * @param
		 * @return void
		 */
		private function fault(event:FaultEvent):void
		{
			facade.sendNotification(MyProfileFacade.LOAD_POST_FAILED, this);
		}
		
		
		
		
		
		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Data Parsing functions
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 전달된 result 객체를 분석하여 Value Object를 생성
		 * 
		 * @param
		 * @return 인사기록 Value Object
		 */
		private function parsePost(retObj:ArrayCollection):ArrayCollection
		{
			var arr:ArrayCollection = new ArrayCollection();
			for each(var rec:Object in retObj)
			{
				var vo:PostVO = new PostVO();
				vo.PSTLZ		= rec["PSTLZ"];
				vo.PFULL		= rec["PFULL"];
				arr.addItem(vo);
			}
			return arr;
		}
	}
}